
What You Need

Obviously you’re going to need IBM VisualAge PL/I for Windows with the latest fixpak, since that’s where you get the headers and the PL/I language features the headers use. 

You’ll also need a Microsoft Platform SDK, which you can download from Microsoft’s developer web site (http://www.microsoft.com/msdownload/platformsdk/sdkupdate) for free (well, actually the phone bill would be quite hefty in my case, but the CD can be ordered from that page for the cost of shipping) - as this is where you’ll find the documentation and the LIB files that you need for linking.

You’ll also need a supported operating system - Windows NT4 SP6a, Windows 2000, Windows XP or Windows 2003. Applications you develop may run on Windows 98 and Windows ME, if you take extra care with the coding to cope with the fact that many Windows APIs are unavailable or have serious functional restrictions.
Finally, you need to know something about coding Windows applications using the procedural API. There are books and courses on this; most of them use C as the sample language. This article assumes you know how to code Windows applications.
What’s Supplied
Header files are provided for the base API, core security services, (access control, LSA, SSPI), windowing engine, GDI, spooler, common controls (including all of the macro APIs) and dialogues, themes, registry, OpenGL, core multimedia, basic shell, simple MAPI, CMC and fax, message queuing, network DDE, sockets, ODBC, windows network management (previously known as LAN Manager), RPC (including the marshalling utility routines), performance data, setup and installation (MSI).
In addition, header files are supplied for some important non-Windows products such as IBM DB2 (unfortunately these are somewhat old and I haven’t had time to update them), OO-REXX and IPF (for those porting OS/2 IPF documentation).

There are also headers for the X11 windowing toolkit for use with third party X-windows managers, and for Tcl/Tk.
To assist OpenGL programmers, both the AUX and the GLUT toolkits are provided as DLLs with suitable LIBs. There are also small sample programs illustrating the use of both toolkits.

Finally, for SNA integrators, the CPI-C API is supported.

The explanation for what is supplied is closely tied to my requirements and specific requests from colleagues of the “I need access to the such-and-such API” sort.

What’s Not Supplied

First off, as already mentioned, LIB files and API documentation are not included - rather you have to obtain a copy of the Microsoft Platform SDK. 

The most serious restriction is that there’s nothing COM-related. For Windows, this means an awful lot of APIs are not accessible.
Not only is the IBM VA PL/I compiler’s OO-PL/I support not officially documented, more seriously the compiler doesn’t use a C++-compatible vtable in the code it does generate. As a result it is not possible to use that support to create “sensible” bindings for COM objects. Rather they would need to be simulated - rather like the C bindings that one occasionally sees in books about COM, but hardly encounters in reality. Additionally, most C++ headers for COM-based APIs are themselves generated monstrosities that skip a lot of the original semantic, so starting from these is rather pointless. 

The original source is the Microsoft Interface Definition Language - a Microsoft COM-enhanced version of the RPC IDL as standardized by DCE - the C++ headers are produced by running this through the MIDL compiler. Basically, somebody would need to write a MIDL compiler for PL/I to create the original headers.
That said, the header files do include the RPC APIs, including marshalling routines, so that MIDL-generated stubs could be executed.

So, once the PL/I OO support is available, the main issue is that of writing a MIDL compiler generating PL/I code. Since MIDL includes language features that exceed the capabilities of C++, such as ref or size_is - that have native PL/I counterparts - the generated PL/I code can retain much of the semantic of the original that is lost in the C++ version.
Also omitted is GDI+. GDI+ is essentially a massive C++ class library, and here, too, OO-PL/I is not officially documented.

Not all of the security APIs are currently available. Omissions, that may be rectified in a later release, include the authz high-level APIs, certificate and credentials management, CryptoAPI and smart card support.
If you’re coding for Windows 95, Windows 98 or Windows ME, you cannot use Unicode as there’s no support for the Microsoft Layer for Unicode. There is also no support for any DOS-based platform-specific features (DOS interrupt APIs, thunking, Device I/O, volume locking, etc). My stock answer to all questions will be that these platforms are not supported, and you should be using a real Windows operating system.
A Short Sample Program

The following short program uses the ShellExecute API to put up the shell search dialogue, starting at the directory given on the command line, or the current directory if no starting directory is specified.

 swffndfl: Procedure (hInstance,

                      hPrevInstance,

                      pszCmdLine,

                      iCmdShow)

            options(winmain reentrant) reorder;

 dcl (length, maxlength, null, stg, sysnull) builtin;

 %include winbase;

 %include shellapi;

 dcl hInstance     type HINSTANCE byvalue,

     hPrevInstance type HINSTANCE byvalue,

     pszCmdLine    pointer byvalue, /* to nul-terminated string */

     iCmdShow      fixed bin (31) byvalue;

 dcl curdir char (MAX_PATH) varz init ('');

 if pszCmdLine ^= sysnull() then

   curdir = pli_p2z (pszCmdLine);

 if curdir = '' then

   call GetCurrentDirectory(stg(curdir), curdir);

 call ShellExecute (sysnull(), "find", curdir, *, *, 0);

 End swffndfl;

To get a clean compilation, you’ll need to specify LIMITS(EXTNAME(100) FIXEDBIN(31,63) FIXEDDEC(31) NAME(100)) PP(MACRO).
The first thing to note is the use of options(winmain) to declare the program as being a Windows GUI application with the standard parameter set.
The reason the winmain entry receives a pointer byvalue, rather than a char(*) varz byaddr is that, in PL/I, varying length strings need a descriptor to support the maxlength and size builtin functions, and these are not available from the Windows. The use of pli_p2z to extract the string is discussed below.

The * in the call to ShellExecute means to omit the optional parameter. The omission of an optional parameter is actually indicated by passing sysnull() as its address, which is why the optional attribute is only allowed for byaddr parameters (or byvalue entry parameters).

The program shown above is actually used to implement the “Find Files” menu command in my Workframe - Visual Source Safe integration package that provides for PL/I source code management in Visual Source Safe, as well as fully automated builds of complex projects - including a colleague’s Perl-based makemake utility for PL/I programs. The package is available at no cost on an as-is basis; there is documentation on its usage and target environment customization.

Introduction to Macro APIs

The pli_p2z macro is used to dereference the pointer that we are passed into a string that we can process. It can be found, together with a bunch of other useful PL/I-specific macros, in windef.cpy (which you should consult). Its declaration is:

 dcl PLI_STRINGA  char(32767) varz based,

     PLI_STRINGW WCHAR(16383) varz based;

%dcl PLI_STRING char ext;

 %if UNICODE ^= '' %then %do;

   %PLI_STRING = 'PLI_STRINGW';

 %end; %else %do;

   %PLI_STRING = 'PLI_STRINGA';

 %end; 

 %dcl pli_p2z entry;  /* ptr -> [(*) ptr -> ] Generic string byaddr*/

 %pli_p2z: proc (p, i) returns (char);

   dcl p char;    /* pointer to TCHAR(*) varz or (*) ptr thereof */

   dcl i char;    /* array subscript, counting from 1, if (*) ptr */

   if parmset(i) then

     return ('ptradd (' || p || ' -> PLI_POINTER , '

                        || '(i - 1) * stg (PLI_POINTER))' ||

             ' -> PLI_STRING ');

   else

     return (p || ' -> PLI_STRING ');

 %end pli_p2z;

User Interface Programming and WindowsX.
Anybody who has ever tried to code a window procedure using the basic Windows API will have discovered that cracking the messages is the single largest source of coding errors - leading to unfortunate results. MS Visual C++ programmers have MFC to help them with basic window procedures and C programmers “in the know” have had access to a poorly documented macro collection known as WindowsX (see windowsx.h). Many other C programmers have never looked into WindowsX.
This collection of macros is also available to PL/I programmers (windowsx.cpy). In fact, for PL/I programmers there is a bonus in the form of a HANDLE_DLGMSG macro that provides the same support for dialogue windows that HANDLE_MSG does for normal window procedures.
Here is a sample dialogue window procedure, and one of the smaller routines, from one of my programs: 

 wp_ycpmdlg: proc (hwnd, msg, wParam, lParam)

             returns (type BOOL byvalue optional) CALLBACK;

   dcl hwnd   type HWND,

       msg    type UINT,

       wParam type WPARAM,

       lParam type LPARAM;

   select (msg);

     HANDLE_DLGMSG (hwnd, WM_INITDIALOG, ycpmdlg_OnInitDialog);

     HANDLE_DLGMSG (hwnd, WM_COMMAND,    ycpmdlg_OnCommand);

     HANDLE_DLGMSG (hwnd, WM_DROPFILES,  ycpmdlg_OnDropfiles);

     HANDLE_DLGMSG (hwnd, WM_CLOSE,      ycpmdlg_OnClose);

     HANDLE_DLGMSG (hwnd, WM_DESTROY,    ycpmdlg_OnDestroy);

     otherwise ;

   end; /* select */

   return (FALSE);

 end wp_ycpmdlg;

 ycpmdlg_OnDropfiles: proc (hwnd, hdrop);

   dcl hwnd  type HWND  byvalue,

       hdrop type HDROP byvalue;

   dcl szFileName char (MAX_PATH - 1) varz;

   call DragQueryFile (hdrop, 0, szFileName, maxlength (szFileName));

   call DragFinish (hdrop);

   call SetDlgItemText (hwnd, EN_PROMPT, szFileName);

 end ycpmdlg_OnDropfiles;

Within ycpmdlg_OnInitDialog, the DragAcceptFiles Shell API is called to enable file dropping (it is also called in the ycpmdlg_OnDestroy to disable file dropping, of course). When a file is dropped, the Windows shell sends a WM_DROPFILES message to the enabled window. In this case, my dialogue procedure retrieves the the name of the (first) dragged file and places it into the EN_PROMPT entry field.
To understand how to use WindowsX, you need to look into it. For the WM_DROPFILES message, WindowsX contains:

/* dcl Cls_OnDropFiles entry (type HWND, type HDROP) */

 HANDLE_WM_DROPFILES: proc (hwnd, wParam, lParam, fn)

                      returns (type LRESULT byvalue optional)

                      options (inline reorder) internal;

   dcl hwnd   type HWND byvalue,

       wParam type WPARAM byvalue,

       lParam type LPARAM byvalue,

       fn     entry (type HWND byvalue, type HDROP byvalue);

   dcl ptrvalue builtin;

   call fn (hwnd, ptrvalue (wParam));

   return (0);

 end HANDLE_WM_DROPFILES;

 FORWARD_WM_DROPFILES: proc (hwnd, hdrop, fn)

                       options (inline reorder) internal;

   dcl hwnd  type HWND byvalue,

       hdrop type HDROP byvalue,

       fn    type WNDPROC;              /* SendMessage or PostMessage */

   dcl binvalue builtin;

   call fn (hwnd, WM_DROPFILES, binvalue (hdrop), 0);

 end FORWARD_WM_DROPFILES;

The first line - the comment, tells you how to declare your message handling routine; the declaration is also used in the “dcl fn” parameter, so the compiler will issue a message if you get it wrong. The window procedure is free to name the HWND as it wishes, as this is a parameter to the HANDLE_* macro. However, the window parameters must be named wParam and lParam. The HANDLE_WM_DROPFILES macro is used internally by the HANDLE_DLGMSG message, you never need to write it.
The FORWARD_* messages are there to permit you to do the opposite of the HANDLE_* messages - pack up the parameters into WPARAM and LPARAM, and send or post them off to some window. As such, they are essentially macros to send specific messages, that conveniently hide the details of the parameter packing.

Window Procedures

Window procedures have a well-defined signature, which can be found in winuser.cpy:

 define alias DLGPROC

              entry (type HWND,

                     type UINT,

                     type WPARAM,

                     type LPARAM)

              returns (optional type BOOL byvalue)

              limited CALLBACK;

You’re window procedure must be declared accordingly; if you make a mistake and you specify RULES(NOLAXLINK) when you compile, the compiler will complain when you try to register it. The default is RULES(LAXLINK), which means that mismatching calling conventions will not be flagged at compile time, rather your program will do funny things at run-time. You’ll find that setting RULES (NOLAXCTL NOLAXDCL NOLAXIF NOLAXLINK NOMULTICLOSE) can be rather helpful in avoiding careless mistakes. I have received more than one email about programs failing, where the author had messed up the declaration and specified LAXLINK to prevent the compiler’s complaining!
The CALLBACK macro (windef.cpy) is:

 %CALLBACK   = 'options (byvalue linkage(stdcall) nodescriptor)';

By contrast, the message crackers do not use stdcall linkage convention, but retain the user’s preferred PL/I default linkage.

More About Macro APIs

As already mentioned The FORWARD_* messages are not really forwarders, so much as macros to send specific messages. As Microsoft developed the common controls, they enhanced the status of message forwarders, and gave them real API names. For example Animate_Open, which I use to set up one of those cute “I’m busy” animations:

   call Animate_Open (hwndAVI, pli_p2z (MAKEINTRESOURCE (AVI_WAIT)));

is a documented API that is actually just a macro that sends the ACM_PLAY message to the specified animation control (commctrl.cpy):
Animate_Open: proc (hwnd, szName)

               returns (type BOOL byvalue optional)

               options (inline) internal;

   dcl hwnd   type HWND byvalue,

       szName TCHAR(*) varz byaddr optional;

   dcl (addr, binvalue) builtin;

   return (SNDMSG (hwnd, ACM_OPEN, 0, binvalue (addr (szName))));

 end Animate_Open;

The call to Animate_Open illustrates another PL/I feature. The second parameter in the PL/I declaration is for a char (*) varz byaddr or wchar (*) varz byaddr (TCHAR is a macro that expands to char or wchar, depending upon whether the UNICODE precompiler variable has been set). The original C declaration used type LPTSTR. But since PL/I considers strings to be a basic type, the PL/I header files uses byaddr strings rather than byvalue pointers as the translation of LP*STR parameters. In fact it uses byaddr rather than pointers or handles wherever that makes sense and would be permitted; for example, they would not permitted for strings in callbacks, for the reason discussed above.
In this case, however, Microsoft have overloaded the string address with a binary resource number: if the address is less than 216, it represents the numeric id of a resource that has been linked into the executable as an address. In our case, we have done this - we actually want to pass that numeric id - so we need to somehow fool the compiler that we know what we’re doing. MAKEINTRESOURCE creates a pointer (this is a standard documented macro API from winuser.cpy), and the pli_p2z macro will pun this to a TCHAR (*) varz, as we saw earlier. This trick persuades the compiler to accept the pointer as a string reference, and since it is being passed byaddr to a nodescriptor routine, it is not dereferenced. The result is compatible with the C style, and is just as grubby.

Important C Compatibility Features in PL/I

There are a few areas in which the PL/I compiler helps with C compatibility. The CAST type function is the most important such feature - it provides C-compatible casting. Another, less-well-known feature is the ability to pass scalars to nodescriptor entries, despite the declaration’s specifying that an array is expected. Consider the following API (wingdi.cpy):

dcl DPtoLP

     entry (type HDC,

            (*) type POINT byaddr,

            fixed bin(31))

     returns (type BOOL byvalue optional)

     WINGDIAPI WINAPI external('DPtoLP');

The function maps device coordinates into logical coordinates (there is also an LPtoDP). The second parameter is defined as the address of an array of POINTs, the number of POINTs in the array being specified by the third parameter. In C, as far as this API is concerned, a single POINT and an array thereof are equivalent - the second parameter is defined as an LPPOINT - the address of the (first) POINT. In PL/I, arrays and scalars are not the same, but for C compatibility, the compiler will accept a scalar POINT in a call, provided the procedure is declared as nodescriptor, as is the case here (see the WINAPI macro). A warning will be issued to this effect; you can use the compiler exit to convert this into an informational message and/or suppress it entirely.

